<commit_before>{ stdenv, fetchurl }:

stdenv.mkDerivation rec {
  name = "icmake-${version}";
  version = "7.22.01";

  src = fetchurl {
    url = "mirror://sourceforge/icmake/icmake_${version}.orig.tar.gz";
    sha256 = "1iv6p9cyvr9i2sjhklplr65llg1ycxqy7z4dfgn0nkwxgs9yf8mm";
  };

  preConfigure = ''
    patchShebangs ./
    sed -i "s;usr/;;g" INSTALL.im
  '';

  buildPhase = "./icm_bootstrap $out";

  installPhase = "./icm_install all /";

  meta = with stdenv.lib; {
    description = "A program maintenance (make) utility using a C-like grammar";
    homepage = http://icmake.sourceforge.net/;
    license = licenses.gpl3;
    maintainers = with maintainers; [ pSub ];
    platforms = platforms.linux;
  };
}<commit_msg>icmake: Sourceforge -> (fetchFrom)GitHub<commit_after>{ stdenv, fetchFromGitHub }:

let version = "7.22.01"; in
stdenv.mkDerivation {
  name = "icmake-${version}";

  src = fetchFromGitHub {
    sha256 = "1pgl8bami4v86ja40in4fsdx940f6q85l1s4b9k53zl29pm85v5k";
    rev = version;
    repo = "icmake";
    owner = "fbb-git";
  };

  sourceRoot = "icmake-${version}-src/icmake";

  preConfigure = ''
    patchShebangs ./
    substituteInPlace INSTALL.im --replace "usr/" ""
  '';

  buildPhase = ''
    ./icm_bootstrap $out
  '';

  installPhase = ''
    ./icm_install all /
  '';

  meta = with stdenv.lib; {
    inherit version;
    description = "A program maintenance (make) utility using a C-like grammar";
    homepage = https://fbb-git.github.io/icmake/;
    license = licenses.gpl3;
    maintainers = with maintainers; [ nckx pSub ];
    platforms = platforms.linux;
  };
}
{'fraction_non_alphanumeric': 0.14694894146948942, 'fraction_numerical': 0.03424657534246575, 'mean_word_length': 3.1848958333333335, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 1, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2400229', 'n_tokens_mistral': 638, 'n_tokens_neox': 604, 'n_words': 112}